Drucker mit Dokumenten speichern, Teil 1
Versionen: Word 2003, 2002/XP, 2000 und 97

Wenn Sie im Netzwerk arbeiten, können Sie meistens auf mehrere Drucker zugreifen und müssen bei jedem Dokument zunächst einmal überlegen, auf welchem Drucker es auszugeben ist. In diesem zweiteiligen Tipp der Woche zeigen wir Ihnen, wie Sie Word automatisch den richtigen Drucker für Sie auswählen lassen. So können Sie Drucksachen mit Grafiken oder Bildern auf dem Farbdrucker ausgeben, der Laserdrucker kommt für die normale Korrespondenz zum Einsatz und für Endlos-Etiketten oder Formulare mit Durchschlägen wird eventuell ein Nadeldrucker eingesetzt.

Die Lösung funktioniert wie ein Überwachungsmechanismus: Mit einem Makro ordnen Sie einem Dokument zunächst den Drucker zu, auf dem es ausgedruckt werden soll. Andere Makros überwachen alle Dokumente, die geöffnet und geschlossen werden. Entdecken sie dabei ein Dokument, dem ein Spezialdrucker zugeordnet ist, setzen sie diesen Drucker beim Öffnen als Standarddrucker bzw. heben diese Einstellung beim Schließen wieder auf. Um diese Lösung in Word einzubinden, gehen Sie wie folgt vor:

· Starten Sie Word und wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung.

· Markieren Sie im Projekt-Explorer (ggf. über Strg+R einblenden) den Eintrag "Normal", und wählen Sie das Menü Einfügen-Modul an, um ein neues Modul mit dem Namen "Modul1" einzufügen.

· Im Eigenschaftenfenster (Anzeige per F4) sollten Sie das Modul über das Feld "(Name)" in "modDOCPrinter" umbenennen.

· Geben Sie im allgemeinen Teil des Moduls die folgenden Deklarationen ein:

Option Explicit
Public strActivePrinter As String

Sub AutoClose()

  DOCPrinter_Reset

End Sub

Sub AutoOpen()

  DOCPrinter_Init

End Sub

Sub DOCPrinter_Auswählen()
  Dim dlg As Dialog
  Dim R As Variant
  Dim strPrinterName As String

  Set dlg = Dialogs(wdDialogFilePrintSetup)
  R = dlg.Display
  If R <> -1 Then Exit Sub

  strPrinterName = dlg.Printer
  On Error Resume Next
  ActiveDocument.Variables.Add "DOCPrinter", _
  strPrinterName
  If Err <> 0 Then
    ActiveDocument.Variables("DOCPrinter").Value = _
    strPrinterName
  End If

  Beep
  MsgBox "Drucker für dieses Dokument wurde auf '" & _
          strPrinterName & "' eingestellt..."

End Sub

Sub DOCPrinter_Init()
  Dim strPrinterName As String

  On Error Resume Next
  strPrinterName = ActiveDocument.Variables("DOCPrinter")
  If Err = 0 Then 'Umschalten

'Aktuellen Drucker festhalten
    strActivePrinter = ActivePrinter

'Neu setzen
    ActivePrinter = strPrinterName
    Beep
    MsgBox "Drucker für dieses Dokument wurde auf '" & _
    strPrinterName & "' umgestellt..."
  End If

End Sub

Sub DOCPrinter_Löschen()

  On Error Resume Next
  ActiveDocument.Variables("DOCPrinter").Delete

End Sub

Sub DOCPrinter_Reset()

  If strActivePrinter <> "" Then 'Umschalten

'Aktuellen Drucker zurücksetzen
    ActivePrinter = strActivePrinter
    strActivePrinter = ""
    Beep
    MsgBox "Standarddrucker für Word wurde auf '" & _
    strActivePrinter & "' zurückgesetzt..."
  End If

End Sub
· Klicken Sie auf Speichern und verlassen Sie die VBA-Entwicklungsumgebung.

Damit sind alle notwendigen Makros in Ihr Word eingebunden. Um einem Dokument einen speziellen Drucker zuzuordnen, gehen Sie wie folgt vor:

1. Öffnen Sie das Dokument, dem ein Drucker zugeordnet werden soll, drücken Sie Alt+F8, um die Makroauswahl zu öffnen und wählen Sie das Makro "DOCPrinter_Auswählen" per Doppelklick aus.

2. Es wird daraufhin ein Dialog "Druckereinrichtung" angezeigt. Wählen Sie hier den gewünschten Drucker per Doppelklick aus und klicken Sie im Anschluss auf Schliessen.

3. Eine kleine Erfolgsmeldung informiert Sie abschließend, dass der Drucker dem Dokument zugeordnet wurde.

Wenn Sie in Zukunft dieses Dokument öffnen, wird es daraufhin geprüft, ob ihm ein Spezialdrucker zugeordnet ist. Ist das der Fall, wird dieser Drucker als Standarddrucker gesetzt. Eine kleine Meldung informiert Sie darüber. Beim Ausdruck dieses Dokumentes werden Sie dann feststellen, dass der gewählte Spezialdrucker automatisch als Ziel verwendet wird, egal, was unter Word als Standarddrucker eingetragen ist. Nach der Bearbeitung und dem Ausdruck schließen Sie das Dokument wieder, wobei geprüft wird, ob dem Dokument ein Spezialdrucker zugeordnet war. Wenn ja, wird der zuvor eingestellte und zwischengespeicherte Standarddrucker wiederhergestellt.

In der nächsten Ausgabe zeigen wir Ihnen eine Besonderheit, die Sie beim Einsatz der Lösung beachten müssen und erläutern Ihnen die Funktionsweise der einzelnen VBA-Makros.

Wenn Sie im Netzwerk arbeiten, können Sie meistens auf mehrere Drucker zugreifen und müssen bei jedem Dokument zunächst einmal überlegen, auf welchem Drucker es auszugeben ist. In diesem zweiteiligen Tipp der Woche zeigen wir Ihnen, wie Sie Word automatisch den richtigen Drucker für Sie auswählen lassen.

In der letzten Ausgabe haben Sie bereits das Makro "DOCPrinter_Auswählen" an Ihre Dokumentvorlage gebunden, über das jedem Dokument auf Wunsch ein Spezialdrucker zugewiesen werden kann. Im Folgenden möchten wir Ihnen einige Besonderheiten der Lösung vorstellen und Ihnen zeigen wie die VBA-Makros im Einzelnen funktionieren.

Bitte beachten Sie beim Einsatz dieser Lösung, dass die Dokumente mit Spezialdrucker nicht "verschachtelt" aufgerufen werden können. Warum das so ist, wird am besten an einem Beispiel deutlich: Sie haben als Standarddrucker "ABC-Drucker" eingestellt und öffnen nun ein Dokument, dem der Spezialdrucker "DEF-Drucker" zugeordnet ist. Das führt dazu, dass "ABC-Drucker " als ursprünglicher Standarddrucker festgehalten und der Standarddrucker auf "DEF-Drucker" umgestellt wird. Öffnen Sie nun ein weiteres Dokument mit Spezialdrucker "GHIDrucker ", wird der aktuelle Standarddrucker "DEF-Drucker" zwischengespeichert und "GHI-Drucker" als Standarddrucker gesetzt. Egal, welches Dokument Sie schließen, es wird auf alle Fälle der falsche Standarddrucker zurückgesetzt, da die Informationen zu "ABC-Drucker" überschrieben sind. Außerdem besteht die Gefahr, dass das zuerst geöffnet "DEF-Drucker"-Dokument irrtümlich auf dem als Standard gesetzten "GHIDrucker" des danach geöffneten Dokumentes ausgedruckt wird.

In der Praxis werden Sie sich aber schnell daran gewöhnt haben, Dokumente mit Spezialdrucker in einem Durchgang zu öffnen, zu bearbeiten, zu drucken und direkt wieder zu schließen. Die Lösung besteht aus den drei Makros "DOCPrinter_Auswählen" (Spezialdrucker zuordnen), "DOCPrinter_Init" (Spezialdrucker setzen) sowie "DOCPrinter_Reset" (Standarddruckerzurücksetzen). Außerdem sind noch die Makros "AutoOpen" und "AutoClose" vorhanden, die Word bei jedem Öffnen und Schließen eines Dokumentes aufruft und die so die Verbindung zu "DOCPrinter_Init" und "DOCPrinter_Reset" herstellen. Für die Auswahl des Druckers wird ein interner Word-Dialog wie folgt genutzt:

Sub DOCPrinter_Auswählen()

  Set dlg = Dialogs(wdDialogFilePrintSetup)
  R = dlg.Display
  If R <> -1 Then Exit Sub

Hier erfolgt zunächst die Anzeige des Dialogs "Druckereinrichtung ". Wird ein Drucker ausgewählt und auf Schließen geklickt, ist die Rückmeldung "-1" (OK). Andernfalls wird die Routine ohne weitere Aktionen verlassen.

strPrinterName = dlg.Printer
  On Error Resume Next
  ActiveDocument.Variables.Add "DOCPrinter", _
  strPrinterName
End Sub

Das Ergebnis der Auswahl, also der Drucker- und Anschlussname, wird dann in einer Dokumentvariablen "DOCPrinter" festgehalten und so mit dem Dokument gespeichert. Beim Öffnen eines Dokumentes wird über "AutoOpen" die Prozedur "DOCPrinter_Init" aufgerufen:

Sub DOCPrinter_Init()
  Dim strPrinterName As String

  On Error Resume Next
  strPrinterName = ActiveDocument.Variables("DOCPrinter")
  If Err = 0 Then 'Umschalten

'Aktuellen Drucker festhalten
    strActivePrinter = ActivePrinter 
    ActivePrinter = strPrinterName 'Neu setzen
    Beep
    MsgBox "Drucker für dieses Dokument wurde auf '" & _
           strPrinterName & "' umgestellt..."
  End If

End Sub

Diese Prozedur prüft, ob eine Dokumentvariable "DOCPrinter" vorhanden ist. Sollte das der Fall sein, wird der aktuelle Standarddrucker zwischengespeichert und der in der Dokumentvariablen festgehaltene Drucker gesetzt. "AutoClose" sorgt beim Schließen eines Dokumentes für den Aufruf von "DOCPrinter_Reset":

Sub DOCPrinter_Reset()

  If strActivePrinter <> "" Then 'Umschalten

    'Aktuellen Drucker zurücksetzen
    ActivePrinter = strActivePrinter

    strActivePrinter = ""
    Beep
    MsgBox "Standarddrucker für Word wurde auf '" & _
           strActivePrinter & "' zurückgesetzt..."
  End If

End Sub

